

# NAG Fortran Library Routine Document

## F07PWF (CHPTRI/ZHPTRI)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

### 1 Purpose

F07PWF (CHPTRI/ZHPTRI) computes the inverse of a complex Hermitian indefinite matrix  $A$ , where  $A$  has been factorized by F07PRF (CHPTRF/ZHPTRF), using packed storage.

### 2 Specification

```
SUBROUTINE F07PWF(UPLO, N, AP, IPIV, WORK, INFO)
ENTRY      chptri (UPLO, N, AP, IPIV, WORK, INFO)
INTEGER      N, IPIV(*), INFO
complex      AP(*), WORK(*)
CHARACTER*1   UPLO
```

The ENTRY statement enables the routine to be called by its LAPACK name.

### 3 Description

To compute the inverse of a complex Hermitian indefinite matrix  $A$ , this routine must be preceded by a call to F07PRF (CHPTRF/ZHPTRF), which computes the Bunch–Kaufman factorization of  $A$  using packed storage.

If  $\text{UPLO} = \text{'U}'$ ,  $A = PUDU^H P^T$  and  $A^{-1}$  is computed by solving  $U^H P^T XPU = D^{-1}$  for  $X$ .

If  $\text{UPLO} = \text{'L}'$ ,  $A = PLDL^H P^T$  and  $A^{-1}$  is computed by solving  $L^H P^T XPL = D^{-1}$  for  $X$ .

### 4 References

Du Croz J J and Higham N J (1992) Stability of methods for matrix inversion *IMA J. Numer. Anal.* **12** 1–19

### 5 Parameters

1: UPLO – CHARACTER\*1 *Input*

*On entry:* indicates how  $A$  has been factorized as follows:

if  $\text{UPLO} = \text{'U}'$ ,  $A = PUDU^H P^T$ , where  $U$  is upper triangular;

if  $\text{UPLO} = \text{'L}'$ ,  $A = PLDL^H P^T$ , where  $L$  is lower triangular.

*Constraint:*  $\text{UPLO} = \text{'U}'$  or  $\text{'L}'$ .

2: N – INTEGER *Input*

*On entry:*  $n$ , the order of the matrix  $A$ .

*Constraint:*  $N \geq 0$ .

3: AP(\*) – **complex** array *Input/Output*

**Note:** the dimension of the array AP must be at least  $\max(1, N * (N + 1)/2)$ .

*On entry:* details of the factorization of  $A$  stored in packed form, as returned by F07PRF (CHPTRF/ZHPTRF).

*On exit:* the factorization is overwritten by the  $n$  by  $n$  Hermitian matrix  $A^{-1}$  stored in packed form. More precisely, the  $(i, j)$ th element of  $A^{-1}$  is stored in  $\text{AP}(i + j(j - 1)/2)$  for  $i \leq j$  if  $\text{UPLO} = \text{'U'}$ , and in  $\text{AP}(i + (2n - j)(j - 1)/2)$  for  $i \geq j$  if  $\text{UPLO} = \text{'L'}$ .

4: IPIV(\*) – INTEGER array *Input*

**Note:** the dimension of the array IPIV must be at least  $\max(1, N)$ .

*On entry:* details of the interchanges and the block structure of  $D$ , as returned by F07PRF (CHPTRF/ZHPTRF).

5: WORK(\*) – *complex* array *Workspace*

**Note:** the dimension of the array WORK must be at least  $\max(1, N)$ .

6: INFO – INTEGER *Output*

*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO =  $-i$ , the  $i$ th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO =  $i$ ,  $d_{ii}$  is exactly zero;  $D$  is singular and the inverse of  $A$  cannot be computed.

## 7 Accuracy

The computed inverse  $X$  satisfies a bound of the form

$$|DU^T P^T XPU - I| \leq c(n)\epsilon(|D| |U^T| |P^T| |X| |P| |U| + |D| |D^{-1}|), \text{ if } \text{UPLO} = \text{'U'}, \text{ or}$$

$$|DL^T P^T XPL - I| \leq c(n)\epsilon(|D| |L^T| |P^T| |X| |P| |L| + |D| |D^{-1}|), \text{ if } \text{UPLO} = \text{'L'},$$

where  $c(n)$  is a modest linear function of  $n$ , and  $\epsilon$  is the *machine precision*.

## 8 Further Comments

The total number of real floating-point operations is approximately  $\frac{8}{3}n^3$ .

The real analogue of this routine is F07PJF (SSPTRI/DSPTRI).

## 9 Example

To compute the inverse of the matrix  $A$ , where

$$A = \begin{pmatrix} -1.36 + 0.00i & 1.58 + 0.90i & 2.21 - 0.21i & 3.91 + 1.50i \\ 1.58 - 0.90i & -8.87 + 0.00i & -1.84 - 0.03i & -1.78 + 1.18i \\ 2.21 + 0.21i & -1.84 + 0.03i & -4.63 + 0.00i & 0.11 + 0.11i \\ 3.91 - 1.50i & -1.78 - 1.18i & 0.11 - 0.11i & -1.84 + 0.00i \end{pmatrix}.$$

Here  $A$  is Hermitian indefinite, stored in packed form, and must first be factorized by F07PRF (CHPTRF/ZHPTRF).

## 9.1 Program Text

**Note:** the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      F07PWF Example Program Text
*      Mark 15 Release. NAG Copyright 1991.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER        (NIN=5,NOUT=6)
INTEGER          NMAX
PARAMETER        (NMAX=8)
*      .. Local Scalars ..
INTEGER          I, IFAIL, INFO, J, N
CHARACTER         UPLO
*      .. Local Arrays ..
complex          AP(NMAX*(NMAX+1)/2), WORK(NMAX)
INTEGER          IPIV(NMAX)
CHARACTER         CLABS(1), RLABS(1)
*      .. External Subroutines ..
EXTERNAL          chptrf, chptri, X04DDF
*      .. Executable Statements ..
WRITE (NOUT,*) 'F07PWF Example Program Results'
*      Skip heading in data file
READ (NIN,*) 
READ (NIN,*) N
IF (N.LE.NMAX) THEN
*
*      Read A from data file
*
READ (NIN,*) UPLO
IF (UPLO.EQ.'U') THEN
    READ (NIN,*) ((AP(I+J*(J-1)/2),J=I,N),I=1,N)
ELSE IF (UPLO.EQ.'L') THEN
    READ (NIN,*) ((AP(I+(2*N-J)*(J-1)/2),J=1,I),I=1,N)
END IF
*
*      Factorize A
*
CALL chptrf(UPLO,N,AP,IPIV,INFO)
*
WRITE (NOUT,*) 
IF (INFO.EQ.0) THEN
*
*      Compute inverse of A
*
CALL chptri(UPLO,N,AP,IPIV,WORK,INFO)
*
*      Print inverse
*
IFAIL = 0
CALL X04DDF(UPLO,'Nonunit',N,AP,'Bracketed','F7.4',
+             'Inverse','Integer',RLABS,'Integer',CLABS,80,0,
+             IFAIL)
ELSE
    WRITE (NOUT,*) 'The factor D is singular'
    END IF
END IF
STOP
*
END

```

## 9.2 Program Data

```
F07PWF Example Program Data
 4 :Value of N
'L' :Value of UPLO
(-1.36, 0.00)
( 1.58,-0.90) (-8.87, 0.00)
( 2.21, 0.21) (-1.84, 0.03) (-4.63, 0.00)
( 3.91,-1.50) (-1.78,-1.18) ( 0.11,-0.11) (-1.84, 0.00) :End of matrix A
```

## 9.3 Program Results

F07PWF Example Program Results

Inverse	1	2	3	4
1	( 0.0826, 0.0000)			
2	(-0.0335, 0.0440)	(-0.1408, 0.0000)		
3	( 0.0603,-0.0105)	( 0.0422,-0.0222)	(-0.2007, 0.0000)	
4	( 0.2391,-0.0926)	( 0.0304, 0.0203)	( 0.0982,-0.0635)	( 0.0073,-0.0000)

---